

Krabička z A4

Robert Mařík

27. listopadu 2010




1 Krabička maximálního objemu

Z papíru A4 (210 × 297 mm) vystřihneme v rozích čtverečky, přehneme podle čárkovaných čar a složíme krabičku.

Úkol: udělejte krabičku, která má co největší objem.

```
Sage code
%hide
X=30
A=polygon2d([[X,X],[X,0],[0,0],[0,X]],rgbcolor=(0.5,0.5,0.5))\
+polygon2d([[297-X,X],[297-X,0],[297,0],[297,X]],rgbcolor=(0.5,0.5,0.5))\
+polygon2d([[X,210-X],[X,210],[0,210],[0,210-X]],rgbcolor=(0.5,0.5,0.5))\
+polygon2d([[297-X,210-X],[297-X,210],[297,210],[297,210-X]],rgbcolor=(0.5,0.5,0.5))\
+line([[0,0],[297,0],[297,210],[0,210],[0,0]])\
+line2d([[X,X],[297-X,X],[297-X,210-X],[X,210-X],[X,X]],rgbcolor='black',linestyle='--')
show(A,xmin=0,xmax=297,ymin=0,ymax=210,aspect_ratio=1,axes=False)
```



⁰Podporováno grantem FRVŠ 131/2010.

⁰Dílo je šířeno pod licencí Creative Commons: Uveďte autora – neuzívejte komerčně.

Sage interact code

```
%auto
@interact
def _(X=slider(0, 100, step_size=1, default=30, label='Velikost výřezu')):
    print 'X=', X
    V(x)=x*(210-2*x)*(297-2*x)
    A=polygon2d([[X,X],[X,0],[0,0],[0,X]], rgbcolor=(0.5,0.5,0.5))\
+polygon2d([[297-X,X],[297-X,0],[297,0],[297,X]], rgbcolor=(0.5,0.5,0.5))\
+polygon2d([[X,210-X],[X,210],[0,210],[0,210-X]], rgbcolor=(0.5,0.5,0.5))\
+polygon2d([[297-X,210-X],[297-X,210],[297,210],[297,210-X]], rgbcolor=(0.5,0.5,0.5))\
+line([[0,0],[297,0],[297,210],[0,210],[0,0]])\
+line2d([[X,X],[297-X,X],[297-X,210-X],[X,210-X],[X,X]], rgbcolor='black', linestyle='--')
    show(A, xmin=0, xmax=297, ymin=0, ymax=210, aspect_ratio=1, axes=False)
    # print 'Objem v litrech je ', (V(x=X)/1000000).n()
```

Vyjádříme objem jako funkci délky jedné strany výřezu. Potom hledáme extrém této funkce, tj. zderivujeme a položíme derivaci rovnu nule.

Sage code

```
V(x)=x*(210-2*x)*(297-2*x)
A=solve(diff(V(x),x),x)
A
```

$$\left[x = -\frac{1}{2} \sqrt{7771} + \frac{169}{2}, x = \frac{1}{2} \sqrt{7771} + \frac{169}{2} \right]$$

Derivace má dva nulové body. Aproximujeme je pro lepší představu desetinným číslem a vidíme, že jeden ze stacionárních bodů nereprezentuje reálnou krabíčku. Ten bod, který vyhovuje je tou správnou volbou.

Sage code

```
B=[i.rhs().n() for i in A]
B
```

[\[40.4233621971911, 128.576637802809\]](#)

Dosadíme velikost optimálního výřezu do funkčního předpisu a najdeme optimální objem (v litrech)

Sage code

```
print 'V( ', B[0], ')=' , (V(B[0])/1000000).n()
```

`V(40.4233621971911)= 1.12849510473126`

Pro zajímavost si můžeme nakreslit průběh funkce, udávající jak se mění objem s velikostí výřezů.

Sage code

```
plot(V(x), (x,0,140), zorder=0) + point((B[0],V(B[0])), rgbcolor='red', zorder=1)
```

